Kepler Problem - Trott 3.1-24b 09-07-16

N. T. Gladd

Initialization: Be sure the files *NTGStylesheet2.nb* and *NTGUtilityFunctions.m* is are in the same directory as that from which this notebook was loaded. Then execute the cell immediately below by mousing left on the cell bar to the right of that cell and then typing “shift” + “enter”. Respond “Yes” in response to the query to evaluate initialization cells.

In[116]:=

```
SetDirectory[NotebookDirectory[]];
(* set directory where source files are located *)
SetOptions[EvaluationNotebook[], (* load the StyleSheet *)
 StyleDefinitions → Get["NTGStylesheet2.nb"]];
Get["NTGUtilityFunctions.m"]; (* Load utilities package *)
```

Purpose

In *The Mathematica GuideBook for Symbolics*, Trott presents an interesting application of pattern matching and function definition for the purpose of generating a perturbative solution to a classical problem in celestial mechanics.
https://www.amazon.com/Mathematica-GuideBook-Symbolics-DVD/dp/0387950206/ref=sr_1_5?ie=UTF8&qid=1473115233&sr=8-5&keywords=Michael+Trott

I often have need to manipulate vector expression and wanted to better appreciate Trott’s treatment of this problem. To that end I follow Trott’s treatment but use my own programming style and notation. I also take the opportunity to consider some logical extensions of the topic.

Background

The Kepler problem has played an important historical role in classical mechanics. The equation for the interaction of two gravitating bodies can be written (in suitably scaled coordinates)

$$\frac{d^2 \vec{r}(t)}{dt^2} = -\frac{\vec{r}(t)}{r(t)^3} \tag{1}$$

where r(t) = $\sqrt{\vec{r}(t)\cdot\vec{r}(t)}$.

Derivations of equation (1) are available in most text books on mechanics. A quick Google search turns up the following derivation *http://web.mit.edu/8.01t/www/materials/modules/guide17.pdf*.

In the reference Sconzo, LeSchack, Tobey (1965), mentioned by Trott, which is available at
http://adsabs.harvard.edu/full/1965AJ.....70..269S
some of the history of this problem is discussed. In particular, it is pointed out that the perturbative solution of the Kepler problem can be written

$$\vec{r}(t) = f(t)\,\vec{r}(0) + g(t)\,\frac{d\,\vec{r}(0)}{\mathrm{dt}} = f(t)\,\vec{r}(0) + g(t)\,\vec{v}(0)$$

$$f(t) = \sum_{i=0}^{\infty} f_i \frac{\delta t^i}{i!} \qquad g(t) = \sum_{i=0}^{\infty} g_i \frac{\delta t^i}{i!} \tag{2}$$

As will be seen, the calculation of the coefficients f_i and g_i is computationally intense. Coefficients to order 5 were obtained by Lagrange in 1862. Sconzo, LeSchack, Tobey pointed out that this problem is well-suited for computer symbolic calculation and, in an early application of symbolic computing, published coefficients to order 12. However, there is at least one typographical error in the published coefficients that I will discuss below.

It is useful to compute using a natural notation

In[118]:=
```
<< Notation`
```

In[119]:=
```
Symbolize[ r⃗ ];
```

The series expansion of $\vec{r}(t)$ is

In[120]:=
```
Series[r⃗[t], {t, 0, 5}]
```

Out[120]=

$$\vec{r}[0] + \vec{r}'[0]\,t + \frac{1}{2}\vec{r}''[0]\,t^2 + \frac{1}{6}\vec{r}^{(3)}[0]\,t^3 + \frac{1}{24}\vec{r}^{(4)}[0]\,t^4 + \frac{1}{120}\vec{r}^{(5)}[0]\,t^5 + O[t]^6$$

The task is to express the higher derivatives in terms of $\vec{r}(0)$ and $\vec{v}(0)$ and so obtain the f and g coefficients. The first two terms are just the initial conditions, so immediately $f_0 = 1$, $f_1 = 0$, $g_0 = 0$, $g_1 = 1$. From equation 1) it is also obvious that $f_2 = 1/r[0]^3$ and $g_2 = 0$.

I Brute Force

To introduce the topic, I begin with a brute force calculation of the third order term

The second order term is

In[121]:=
```
w1[1] = D[r⃗[t], {t, 2}] == - r⃗[t]/r[t]^3
```
Out[121]=
$$\vec{r}''[t] == -\frac{\vec{r}[t]}{r[t]^3}$$

and the 3rd order term follows from differentiation

In[122]:=
```
w1[2] = MapEqn[D[#, t] &, w1[1]]
```
Out[122]=
$$\vec{r}^{(3)}[t] == \frac{3\,\vec{r}[t]\,r'[t]}{r[t]^4} - \frac{\vec{r}'[t]}{r[t]^3}$$

An explicit expression is needed for r′[t]. A problem arises in that Mathematica’s built in form for Dot arguments specified as lists.

In[123]:=
```
w1[3] = r[t] == √Dot[{x[t], y[t], z[t]}, {x[t], y[t], z[t]}]
```
Out[123]=
$$r[t] == \sqrt{x[t]^2 + y[t]^2 + z[t]^2}$$

In[124]:=
```
w1[4] = MapEqn[D[#, t] &, w1[3]] // Simplify
```
Out[124]=
$$r'[t] == \left(x[t]\,x'[t] + y[t]\,y'[t] + z[t]\,z'[t]\right) \Big/ \left(\sqrt{x[t]^2 + y[t]^2 + z[t]^2}\right)$$

In[125]:=
```
w1[5] = w1[4] /. 1/√(x[t]^2 + y[t]^2 + z[t]^2) → 1/r[t]
```
Out[125]=
$$r'[t] == \frac{1}{r[t]}\left(x[t]\,x'[t] + y[t]\,y'[t] + z[t]\,z'[t]\right)$$

In[126]:=
```
w1[6] = w1[2] /. (w1[5] // ER)
```
Out[126]=
$$\vec{r}^{(3)}[t] == -\frac{\vec{r}'[t]}{r[t]^3} + \frac{1}{r[t]^5} 3\,\vec{r}[t]\,\left(x[t]\,x'[t] + y[t]\,y'[t] + z[t]\,z'[t]\right)$$

Evaluating at t = 0

In[127]:=
```
w1[7] = w1[6] /. t → 0
```
Out[127]=
$$\vec{r}^{(3)}[0] == -\frac{\vec{r}'[0]}{r[0]^3} + \frac{1}{r[0]^5} 3\,\vec{r}[0]\,\left(x[0]\,x'[0] + y[0]\,y'[0] + z[0]\,z'[0]\right)$$

Parameters are introduced

In[128]:=
```
w1[8] = w1[7] /. (x[0] x′[0] + y[0] y′[0] + z[0] z′[0]) → σ r[0]^2 /. r[0] → μ^(-1/3)
```
Out[128]=
$$\vec{r}^{(3)}[0] == 3\,\mu\,\sigma\,\vec{r}[0] - \mu\,\vec{r}'[0]$$

So, at third order f_3 = 3 μ σ, g_3 = -μ.

However, when using the standard definition of Dot, the calculation becomes more onerous at higher orders.

2 Trott’s approach using function definitions

Trott uses Mathematica capabilities to simplify this calculation considerably.
1) He introduces a replacement function for differentiation D[arg, var] → $\mathcal{D}$[arg, var] where rules will be defined that allow arg to be a vector with a symbolic representation.
2) He introduces a replacement function for the dot product of two vectors Dot[{x1, x2, ...},{y1, y2, ...}] → $\mathcal{D}$ot[vec, vec] where vec denotes a symbolic representation of a vector.
3) He attaches properties to the replacement function that causes them to evaluate in a manner that generates the power series.

The series expansion can be expressed in the following abstract form

In[129]:=
```
seriesTerm[0] = r⃗[t];
seriesTerm[n_] := seriesTerm[n] = Expand[𝒟[seriesTerm[n - 1], t]];
```

In[131]:=
```
w2[1] = Table[seriesTerm[i], {i, 1, 5}] // ColumnForm
```
Out[131]=
```
𝒟[r⃗[t], t]
𝒟[𝒟[r⃗[t], t], t]
𝒟[𝒟[𝒟[r⃗[t], t], t], t]
𝒟[𝒟[𝒟[𝒟[r⃗[t], t], t], t], t]
𝒟[𝒟[𝒟[𝒟[𝒟[r⃗[t], t], t], t], t], t]
```

By attaching specific definitions and rules to $\mathcal{D}$ and $\mathcal{D}$ot, seriesTerm evaluates to specific terms in the expansion.

Begin by defining the second derivative to be the rhs of equation (1).

In[132]:=
$$\mathcal{D}[\mathcal{D}[\vec{r}[t], t], t] = -\frac{\vec{r}[t]}{r[t]^3}$$
Out[132]=
$$-\frac{\vec{r}[t]}{r[t]^3}$$

Define the operator $\mathcal{D}$ to distribute itself over sums

In[133]:=
```
𝒟[sum_Plus, t] := 𝒟[#, t] & /@ sum
```

The following rule pulls terms not depending on t outside the operator.

In[134]:=
```
𝒟[a_ b_, t] /; FreeQ[a, t, ∞] := a 𝒟[b, t]
```

The product rule for differentiation is imposed

In[135]:=
```
𝒟[a_ b_, t] := 𝒟[a, t] b + a 𝒟[b, t]
```

A rule for handling powers is also imposed.

In[136]:=
```
𝒟[a_^n_, t] := n a^(n-1) 𝒟[a, t]
```

Subsequent rules involve 𝒟ot

The Orderless attribute is imposed on 𝒟ot to makes the operator commutative.

In[137]:=
```
SetAttributes[𝒟ot, Orderless]
```

Finally, 𝒟ot is defined

In[138]:=
```
𝒟ot[r⃗[t], r⃗[t]] = r[t]^2;
```

Define a rule for derivatives of r[t]. (This is important in triggering simplifications of expressions)

In[139]:=
```
𝒟[r[t]^n_., t] := n r[t]^(n-1) 𝒟ot[r⃗[t], 𝒟[r⃗[t], t]] / r[t]
```

A rule has to be introduced for the application of 𝒟 to 𝒟ot

In[140]:=
```
𝒟[𝒟ot[a_, b_], t] := 𝒟ot[𝒟[a, t], b] + 𝒟ot[a, 𝒟[b, t]]
```

Scalar quantities should be moved outside the 𝒟ot operator

In[141]:=
```
𝒟ot[a_ b_, c_] /; And[FreeQ[a, r⃗[t]], FreeQ[a, 𝒟[r⃗[t], t]]] := a 𝒟ot[b, c]
```

An explicit rule is defined for the time derivative of dot product `dot[r⃗[t], 𝒟[r⃗[t], t]]^n` (This is important in triggering simplifications of expressions)

In[142]:=
```
𝒟[dot[r⃗[t], 𝒟[r⃗[t], t]]^n_, t] :=
 n dot[r⃗[t], 𝒟[r⃗[t], t]]^(n-1) 𝒟[dot[r⃗[t], 𝒟[r⃗[t], t]], t]
```

In summary, the definitions of 𝒟 and 𝒟ot are

In[143]:=
```
?? 𝒟
```

```
Parallel`𝒟

𝒟[𝒟[r⃗[t], t], t] = -r⃗[t]/r[t]^3

𝒟[sum_Plus, t] := (𝒟[#1, t] &) /@ sum

𝒟[a_ b_, t] /; FreeQ[a, t, ∞] := a 𝒟[b, t]

𝒟[a_ b_, t] := 𝒟[a, t] b + a 𝒟[b, t]

𝒟[dot[r⃗[t], 𝒟[r⃗[t], t]]^n_, t] := n dot[r⃗[t], 𝒟[r⃗[t], t]]^(n-1) 𝒟[dot[r⃗[t], 𝒟[r⃗[t], t]], t]

𝒟[a_^n_, t] := n a^(n-1) 𝒟[a, t]

𝒟[r[t]^n_., t] := n r[t]^(n-1) 𝒟ot[r⃗[t], 𝒟[r⃗[t], t]] / r[t]

𝒟[𝒟ot[a_, b_], t] := 𝒟ot[𝒟[a, t], b] + 𝒟ot[a, 𝒟[b, t]]
```

```
?? 𝒟ot
```

```
Global`𝒟ot

Attributes[𝒟ot] = {Orderless}

𝒟ot[r⃗[t], r⃗[t]] = r[t]^2

𝒟ot[a_ b_, c_] /; FreeQ[a, r⃗[t]] && FreeQ[a, 𝒟[r⃗[t], t]] := a 𝒟ot[b, c]
```

With these definitions and rules in place, the seriesTerms expand into explicit forms.

In[144]:=
```
w2[2] = Table[seriesTerm[i], {i, 0, 5}] // Expand;
w2[2] // ColumnForm
```

Out[145]=
```
r⃗[t]
𝒟[r⃗[t], t]
-r⃗[t]/r[t]^3
-𝒟[r⃗[t],t]/r[t]^3 + 3 r⃗[t] 𝒟ot[r⃗[t],𝒟[r⃗[t],t]]/r[t]^5
-2 r⃗[t]/r[t]^6 + 6 𝒟[r⃗[t],t] 𝒟ot[r⃗[t],𝒟[r⃗[t],t]]/r[t]^5 - 15 r⃗[t] 𝒟ot[r⃗[t],𝒟[r⃗[t],t]]^2/r[t]^7 + 3 r⃗[t] 𝒟ot[𝒟[r⃗[t],t],𝒟[r⃗[t],t]]/r[t]^5
-8 𝒟[r⃗[t],t]/r[t]^6 + 30 r⃗[t] 𝒟ot[r⃗[t],𝒟[r⃗[t],t]]/r[t]^8 - 45 𝒟[r⃗[t],t] 𝒟ot[r⃗[t],𝒟[r⃗[t],t]]^2/r[t]^7 + 105 r⃗[t] 𝒟ot[r⃗[t],𝒟[r⃗[t],t]]^3/r[t]^9 + 9 𝒟[r⃗[·
```

Ala Sconzo, LeSchack, Tobey I introduce simplified forms for the various terms appearing in the expansion

In[146]:=
```
def[μ] = μ == 1/r[t]^3;
def[s] = s == 𝒟ot[r⃗[t], 𝒟[r⃗[t], t]];
def[σ] = σ == s/r[t]^2;
def[ϵ] = ϵ == 𝒟ot[𝒟[r⃗[t], t], 𝒟[r⃗[t], t]]/r[t]^2;
```

In[151]:=
```
w2[3] = w2[2] /. Sol[def[s], 𝒟ot[r⃗[t], 𝒟[r⃗[t], t]]] /.
    Sol[def[ϵ], 𝒟ot[𝒟[r⃗[t], t], 𝒟[r⃗[t], t]]] /. Sol[def[σ], s] /. Sol[def[μ], r[t]]
```

Out[151]=
```
{r⃗[t], 𝒟[r⃗[t], t], -μ r⃗[t], 3 μ σ r⃗[t] - μ 𝒟[r⃗[t], t],
 3 ϵ μ r⃗[t] - 2 μ² r⃗[t] - 15 μ σ² r⃗[t] + 6 μ σ 𝒟[r⃗[t], t], -45 ϵ μ σ r⃗[t] + 30 μ² σ r⃗[t] +
  105 μ σ³ r⃗[t] + 9 ϵ μ 𝒟[r⃗[t], t] - 8 μ² 𝒟[r⃗[t], t] - 45 μ σ² 𝒟[r⃗[t], t]}
```

I implement a function for extracting coefficients at each order

In[155]:=
```
Clear[IndentifyCoefficients];
IndentifyCoefficients[arg_, index_] :=
  Module[{i = index[[1]], frhs, grhs},
   {f_(i-1) == Coefficient[arg, r⃗[t]],
    g_(i-1) == Coefficient[arg, 𝒟[r⃗[t], t]]}];

w2[4] = MapIndexed[IndentifyCoefficients[#1, #2] &, w2[3]]
```

Out[157]=
$$\{\{f_0 == 1,\ g_0 == 0\},\ \{f_1 == 0,\ g_1 == 1\},\ \{f_2 == -\mu,\ g_2 == 0\},\ \{f_3 == 3\,\mu\,\sigma,\ g_3 == -\mu\},\ \{f_4 == 3\,\epsilon\,\mu - 2\,\mu^2 - 15\,\mu\,\sigma^2,\ g_4 == 6\,\mu\,\sigma\},\ \{f_5 == -45\,\epsilon\,\mu\,\sigma + 30\,\mu^2\,\sigma + 105\,\mu\,\sigma^3,\ g_5 == 9\,\epsilon\,\mu - 8\,\mu^2 - 45\,\mu\,\sigma^2\}\}$$

In[158]:=
```
LGrid[w2[4], "f-g coefficients"]
```

Out[158]=

f–g coefficients

$f_0 == 1$	$g_0 == 0$
$f_1 == 0$	$g_1 == 1$
$f_2 == -\mu$	$g_2 == 0$
$f_3 == 3\,\mu\,\sigma$	$g_3 == -\mu$
$f_4 == 3\,\epsilon\,\mu - 2\,\mu^2 - 15\,\mu\,\sigma^2$	$g_4 == 6\,\mu\,\sigma$
$f_5 == -45\,\epsilon\,\mu\,\sigma + 30\,\mu^2\,\sigma + 105\,\mu\,\sigma^3$	$g_5 == 9\,\epsilon\,\mu - 8\,\mu^2 - 45\,\mu\,\sigma^2$

From Sconzo, LeSchack, Tobey

$$f_0 = 1$$
$$f_1 = 0$$
$$f_2 = -\mu$$
$$f_3 = 3\sigma\mu$$
$$f_4 = -15\sigma^2\mu + 3\epsilon\mu + \mu^2$$
$$f_5 = 105\sigma^3\mu - \sigma(45\epsilon\mu + 15\mu^2)$$

$$g_0=0$$
$$g_1=1$$
$$g_2=0$$
$$g_3=-\mu$$
$$g_4=6\sigma\mu$$
$$g_5=-45\sigma^2\mu+9\epsilon\mu+\mu^2$$

There is a discrepancy beginning with f_4! The coefficients of μ^2 differ, with the term -2 μ^2 appearing in the derived term and the term $+\mu^2$ appearing in the SLT table.

Which is correct? In Appendix A, I perform an independent calculation that confirms the result derived in this notebook.

3 Numeric and analytic trajectories

I write a function for the analytical trajectory and compare it against a numerical solution. I calculate coefficients to 7th order

In[162]:=
```
w3[1] = Table[seriesTerm[i], {i, 0, 7}] // Expand;
w3[2] = w3[1] /. Sol[def[s], 𝒟ot[r⃗[t], 𝒟[r⃗[t], t]]] /. Sol[def[ϵ],
      𝒟ot[𝒟[r⃗[t], t], 𝒟[r⃗[t], t]]] /. Sol[def[σ], s] /. Sol[def[μ], r[t]];
w3[3] = MapIndexed[IndentifyCoefficients[#1, #2] &, w3[2]]
```

Out[164]=

$$\{\{f_0 == 1, g_0 == 0\}, \{f_1 == 0, g_1 == 1\}, \{f_2 == -\mu, g_2 == 0\},$$
$$\{f_3 == 3\,\mu\,\sigma, g_3 == -\mu\}, \{f_4 == 3\,\epsilon\,\mu - 2\,\mu^2 - 15\,\mu\,\sigma^2, g_4 == 6\,\mu\,\sigma\},$$
$$\{f_5 == -45\,\epsilon\,\mu\,\sigma + 30\,\mu^2\,\sigma + 105\,\mu\,\sigma^3, g_5 == 9\,\epsilon\,\mu - 8\,\mu^2 - 45\,\mu\,\sigma^2\},$$
$$\{f_6 == -45\,\epsilon^2\,\mu + 66\,\epsilon\,\mu^2 - 22\,\mu^3 + 630\,\epsilon\,\mu\,\sigma^2 - 420\,\mu^2\,\sigma^2 - 945\,\mu\,\sigma^4,$$
$$g_6 == -180\,\epsilon\,\mu\,\sigma + 150\,\mu^2\,\sigma + 420\,\mu\,\sigma^3\},$$
$$\{f_7 == 1575\,\epsilon^2\,\mu\,\sigma - 2268\,\epsilon\,\mu^2\,\sigma + 756\,\mu^3\,\sigma - 9450\,\epsilon\,\mu\,\sigma^3 + 6300\,\mu^2\,\sigma^3 + 10\,395\,\mu\,\sigma^5,$$
$$g_7 == -225\,\epsilon^2\,\mu + 396\,\epsilon\,\mu^2 - 172\,\mu^3 + 3150\,\epsilon\,\mu\,\sigma^2 - 2520\,\mu^2\,\sigma^2 - 4725\,\mu\,\sigma^4\}\}$$

fList and gList

In[165]:=
```
{w3[3]〚All, 1, 2〛, w3[3]〚All, 2, 2〛}
```

Out[165]=

$$\{\{1, 0, -\mu, 3\,\mu\,\sigma, 3\,\epsilon\,\mu - 2\,\mu^2 - 15\,\mu\,\sigma^2, -45\,\epsilon\,\mu\,\sigma + 30\,\mu^2\,\sigma + 105\,\mu\,\sigma^3,$$
$$-45\,\epsilon^2\,\mu + 66\,\epsilon\,\mu^2 - 22\,\mu^3 + 630\,\epsilon\,\mu\,\sigma^2 - 420\,\mu^2\,\sigma^2 - 945\,\mu\,\sigma^4,$$
$$1575\,\epsilon^2\,\mu\,\sigma - 2268\,\epsilon\,\mu^2\,\sigma + 756\,\mu^3\,\sigma - 9450\,\epsilon\,\mu\,\sigma^3 + 6300\,\mu^2\,\sigma^3 + 10\,395\,\mu\,\sigma^5\},$$
$$\{0, 1, 0, -\mu, 6\,\mu\,\sigma, 9\,\epsilon\,\mu - 8\,\mu^2 - 45\,\mu\,\sigma^2, -180\,\epsilon\,\mu\,\sigma + 150\,\mu^2\,\sigma + 420\,\mu\,\sigma^3,$$
$$-225\,\epsilon^2\,\mu + 396\,\epsilon\,\mu^2 - 172\,\mu^3 + 3150\,\epsilon\,\mu\,\sigma^2 - 2520\,\mu^2\,\sigma^2 - 4725\,\mu\,\sigma^4\}\}$$

In[166]:=
```
Clear[TrajectoryAnalytic];
TrajectoryAnalytic[{x0_, y0_, z0_}, {vx0_, vy0_, vz0_}, t_, nTerms_] :=
 Module[{rvec0, vvec0, r0, v0, s0, sdot0, μ, σ, ϵ, fList, gList, position},
  rvec0 = {x0, y0, z0};
  vvec0 = {vx0, vy0, vz0};
  r0 = Dot[rvec0, rvec0];
  s0 = Dot[rvec0, vvec0];
  sdot0 = Dot[vvec0, vvec0];
  μ = 1/r0^3;
  σ = s0/r0^2;
  ϵ = sdot0/r0^2;
  fList = {1, 0, -μ, 3 μ σ, 3 ϵ μ - 2 μ^2 - 15 μ σ^2,
    -45 ϵ μ σ + 30 μ^2 σ + 105 μ σ^3, -45 ϵ^2 μ + 66 ϵ μ^2 - 22 μ^3 + 630 ϵ μ σ^2 - 420 μ^2 σ^2 - 945 μ σ^4,
    1575 ϵ^2 μ σ - 2268 ϵ μ^2 σ + 756 μ^3 σ - 9450 ϵ μ σ^3 + 6300 μ^2 σ^3 + 10395 μ σ^5};
  gList = {0, 1, 0, -μ, 6 μ σ, 9 ϵ μ - 8 μ^2 - 45 μ σ^2, -180 ϵ μ σ + 150 μ^2 σ + 420 μ σ^3,
    -225 ϵ^2 μ + 396 ϵ μ^2 - 172 μ^3 + 3150 ϵ μ σ^2 - 2520 μ^2 σ^2 - 4725 μ σ^4};
  position = rvec0 Sum[fList[[i + 1]] t^i/i!, {i, 0, nTerms}] +
    vvec0 Sum[gList[[i + 1]] t^i/i!, {i, 0, nTerms}]]
```

Visualization of solution

In[168]:=

```
Module[{x0 = 1, vx0 = 0.25, vy0 = 1, vz0 = 0.25, tMax = 5,
  nTerms = 3, eqns, initialConditions, trajectory, refCircle, G},
 eqns = {x''[t] == -(x[t] / (x[t]^2 + y[t]^2 + z[t]^2)^(3/2)), y''[t] ==
    -(y[t] / (x[t]^2 + y[t]^2 + z[t]^2)^(3/2)), z''[t] == -(z[t] / (x[t]^2 + y[t]^2 + z[t]^2)^(3/2))};
 initialConditions = {x[0] == x0, y[0] == 0, z[0] == 0,
   x'[0] == vx0, y'[0] == vy0, z'[0] == vz0};
 trajectory = NDSolve[Join[eqns, initialConditions],
   {x[t], y[t], z[t]}, {t, 0, tMax}];
 G[1] = ParametricPlot3D[{{x[t], y[t], z[t]} /. trajectory,
    TrajectoryAnalytic[{x0, 0, 0}, {vx0, vy0, vz0}, t, 4],
    TrajectoryAnalytic[{x0, 0, 0}, {vx0, vy0, vz0}, t, 5],
    TrajectoryAnalytic[{x0, 0, 0}, {vx0, vy0, vz0}, t, 6],
    TrajectoryAnalytic[{x0, 0, 0}, {vx0, vy0, vz0}, t, 7]}, {t, 0, tMax},
   PlotStyle → {Black, Blue, Green, Orange, Red}, AxesLabel →
    {Stl["x"], Stl["y"], Stl["z"]}, PlotRange → {{-2, 2}, {-2, 2}, {-2, 2}},
   PlotLegends → {"numerical", "n = 4", "n = 5", "n = 6", "n = 7"},
   PlotLabel → Stl["Comparison of numerical trajectory with
power series approximation"]];
 refCircle = Line@Table[{Cos[θ], Sin[θ], 0}, {θ, 0, 2 π, π / 50}];
 G[2] = Graphics3D[{PointSize[0.02],
    {Red, Point[{0, 0, 0}], {Black, Point[{x0, 0, 0}]}}, refCircle}];
 Show[
  G[
   1],
  G[
   2]]]
```

Out[168]=



A Analyzing the f_4 discrepancy with Sconzo, LeSchack, Tobey

As pointed out in Section 2, there is a discrepancy between the f_4 terms calculated using $\mathcal{D}$ and $\mathcal{D}$ot operators, and the equivalent term appearing in Sconzo, LeSchack, Tobey 1965. The latter involved the use of recursion relations occurring between terms appearing in the expansion.

$$f_4 = 3\,\epsilon\,\mu - 2\,\mu^2 - 15\,\mu\,\sigma^2$$
$$g_4 = 6\,\sigma\,\mu$$

$$f_4^{\text{SLT}} = 3\,\epsilon\,\mu + \mu^2 - 15\,\mu\,\sigma^2$$
$$g_4^{\text{SLT}} = 6\,\sigma\,\mu$$

I will resolve this discrepancy by making a third calculation of the

$$\vec{r}^{\,(4)}[0] = f_4\,\vec{r}[0] + g_4\,\vec{r}^{\,(1)}[0] \quad (3)$$

and then making numerical comparisons for arbitrary initial conditions.

I will just make a brute force calculation of the second derivative of the rhs of equation (1) without concerning myself with the appearance of the expression. For t = 0, it will be a complicated expression of the initial conditions that, nonetheless, should be identical to the simpler forms given in equation 3.

For the purpose of comparing the different calculations, I choose $\vec{r}[0]$ = {1, 0, 0} but choose the initial velocities to be randomly distributed. The columns are

v0 - randomly chosen initial velocity

d4r - $\vec{r}^{(4)}[0]$ calculated using the recursion of Section 2

d4rSLT - $\vec{r}^{(4)}[0]$ from Sconzo, LeSchack, Tobey 1965

d4rBruteForce - $\vec{r}^{(4)}[0]$ brute force calculation below

d4r - d4rBF - difference between recursion and brute force (should always be small)

In[182]:=
```
Module[{x0 = 1, y0 = 0, z0 = 0, results, info},
 results =
  Table[With[{vx0 = RandomReal[-1, 1][[1]] ,
     vy0 = RandomReal[-1, 1][[1]] , vz0 = RandomReal[-1, 1][[1]] },
    Join[{{vx0, vy0, vz0}}, CompareCalculations[x0, y0, z0, vx0, vy0, vz0]]], {5}];
 info = PrependTo[results, {"v0", "d4r", "d4rSLT", "d4rBruteForce", "d4r - d4fBF"}];
 LGrid[results, "test"]]
```

Out[182]=

test

v0	d4r	d4rSLT	d4rBruteForce	d4r – d4fBF
{−0.494482, −0.123496, −0.916912}	{−0.899141, 0.366398, 2.72038}	{2.10086, 0.366398, 2.72038}	{−0.899141, 0.366398, 2.72038}	{2.22045×10^{-16}, 0., 0.}
{−0.820586, −0.918851, −0.316291}	{−3.20718, 4.52398, 1.55726}	{−0.207183, 4.52398, 1.55726}	{−3.20718, 4.52398, 1.55726}	{-4.44089×10^{-16}, 0., 0.}
{−0.400173, −0.00979359, −0.699461}	{−1.49281, 0.0235148, 1.67943}	{1.50719, 0.0235148, 1.67943}	{−1.49281, 0.0235148, 1.67943}	{0., 0., 0.}
{−0.0206861, −0.918088, −0.113929}	{0.565031, 0.11395, 0.0141405}	{3.56503, 0.11395, 0.0141405}	{0.565031, 0.11395, 0.0141405}	{0., 1.38778×10^{-17}, 0.}
{−0.703682, −0.864678, −0.0105074}	{−2.72767, 3.65075, 0.0443631}	{0.272326, 3.65075, 0.0443631}	{−2.72767, 3.65075, 0.0443631}	{-4.44089×10^{-16}, -4.44089×10^{-16}, 0.}

I conclude that some error inadvertently crept into the recursive calculation of Sconzo, LeSchack, Tobey 1965.

Calculation of $\vec{r}^{(4)}[0]$ and implementation of comparison formulas.

In[177]:=

$$\text{wA}[1] = \text{D}\left[\vec{r}[t], \{t, 2\}\right] == -\frac{\vec{r}[t]}{r[t]^3}$$

Out[177]=

$$\vec{r}''[t] == -\frac{\vec{r}[t]}{r[t]^3}$$

but write this as

In[178]:=
```
wA[2] = D[{x[t], y[t], z[t]}, {t, 2}] ==
  -({x[t], y[t], z[t]} / Dot[{x[t], y[t], z[t]}, {x[t], y[t], z[t]}]^(3/2))
```

Out[178]=

$$\{x''[t], y''[t], z''[t]\} == \left\{-\left(x[t] \Big/ \left(x[t]^2 + y[t]^2 + z[t]^2\right)^{3/2}\right), -\left(y[t] \Big/ \left(x[t]^2 + y[t]^2 + z[t]^2\right)^{3/2}\right), -\left(z[t] \Big/ \left(x[t]^2 + y[t]^2 + z[t]^2\right)^{3/2}\right)\right\}$$

I calculate $\vec{r}^{(4)}[t]$ by differentiating the rhs of wA[2] twice but then use wA[2] as a rule to remove second derivatives

In[179]:=
```
Thread[Rule[wA[2][[1]], wA[2][[2]]]]
```

Out[179]=

$$\left\{x''[t] \to -\left(x[t] \Big/ \left(x[t]^2 + y[t]^2 + z[t]^2\right)^{3/2}\right), y''[t] \to -\left(y[t] \Big/ \left(x[t]^2 + y[t]^2 + z[t]^2\right)^{3/2}\right), z''[t] \to -\left(z[t] \Big/ \left(x[t]^2 + y[t]^2 + z[t]^2\right)^{3/2}\right)\right\}$$

In[180]:=
```
wA[4] = D[wA[2][[2]], {t, 2}] /. Thread[Rule[wA[2][[1]], wA[2][[2]]]] // Simplify
```

Out[180]=

$$\left\{\frac{1}{4\left(x[t]^2 + y[t]^2 + z[t]^2\right)^4}\Big(4\,x[t]\left(x[t]^2 + y[t]^2 + z[t]^2\right) + 24\left(x[t]^2 + y[t]^2 + z[t]^2\right)^{3/2} x'[t]\left(x[t]\,x'[t] + y[t]\,y'[t] + z[t]\,z'[t]\right) - 3\,x[t]\Big(20\sqrt{\left(x[t]^2 + y[t]^2 + z[t]^2\right)}\left(x[t]\,x'[t] + y[t]\,y'[t] + z[t]\,z'[t]\right)^2 - 4\left(x[t]^2 + y[t]^2 + z[t]^2\right)\Big(-1 + \sqrt{\left(x[t]^2 + y[t]^2 + z[t]^2\right)}\,x'[t]^2 + \sqrt{\left(x[t]^2 + y[t]^2 + z[t]^2\right)}\,y'[t]^2 + \sqrt{\left(x[t]^2 + y[t]^2 + z[t]^2\right)}\,z'[t]^2\Big)\Big)\Big),\right.$$

$$\frac{1}{4\left(x[t]^2 + y[t]^2 + z[t]^2\right)^4}\Big(4\,y[t]\left(x[t]^2 + y[t]^2 + z[t]^2\right) + 24\left(x[t]^2 + y[t]^2 + z[t]^2\right)^{3/2} y'[t]\left(x[t]\,x'[t] + y[t]\,y'[t] + z[t]\,z'[t]\right) - 3\,y[t]\Big(20\sqrt{\left(x[t]^2 + y[t]^2 + z[t]^2\right)}\left(x[t]\,x'[t] + y[t]\,y'[t] + z[t]\,z'[t]\right)^2 - 4\left(x[t]^2 + y[t]^2 + z[t]^2\right)\Big(-1 + \sqrt{\left(x[t]^2 + y[t]^2 + z[t]^2\right)}\,x'[t]^2 + \sqrt{\left(x[t]^2 + y[t]^2 + z[t]^2\right)}\,y'[t]^2 + \sqrt{\left(x[t]^2 + y[t]^2 + z[t]^2\right)}\,z'[t]^2\Big)\Big)\Big),$$

$$\left.\frac{1}{4\left(x[t]^2 + y[t]^2 + z[t]^2\right)^4}\Big(4\,z[t]\left(x[t]^2 + y[t]^2 + z[t]^2\right) + 24\left(x[t]^2 + y[t]^2 + z[t]^2\right)^{3/2} z'[t]\left(x[t]\,x'[t] + y[t]\,y'[t] + z[t]\,z'[t]\right) - 3\,z[t]\Big(20\sqrt{\left(x[t]^2 + y[t]^2 + z[t]^2\right)}\left(x[t]\,x'[t] + y[t]\,y'[t] + z[t]\,z'[t]\right)^2 - 4\left(x[t]^2 + y[t]^2 + z[t]^2\right)\Big(-1 + \sqrt{\left(x[t]^2 + y[t]^2 + z[t]^2\right)}\,x'[t]^2 + \sqrt{\left(x[t]^2 + y[t]^2 + z[t]^2\right)}\,y'[t]^2 + \sqrt{\left(x[t]^2 + y[t]^2 + z[t]^2\right)}\,z'[t]^2\Big)\Big)\Big)\right\}$$

In[181]:=
```
wA[5] = wA[4] /. t → 0 /.
  {x[0] → x0, y[0] → y0, z[0] → z0, x′[0] → vx0, y′[0] → vy0, z′[0] → vz0}
```

Out[181]=

$$\Big\{\frac{1}{4\left(x0^2+y0^2+z0^2\right)^4}\Big(4\,x0\left(x0^2+y0^2+z0^2\right)+24\,vx0\left(vx0\,x0+vy0\,y0+vz0\,z0\right)\left(x0^2+y0^2+z0^2\right)^{3/2}-3\,x0\Big(20\left(vx0\,x0+vy0\,y0+vz0\,z0\right)^2\sqrt{x0^2+y0^2+z0^2}-4\left(x0^2+y0^2+z0^2\right)\left(-1+vx0^2\sqrt{x0^2+y0^2+z0^2}+vy0^2\sqrt{x0^2+y0^2+z0^2}+vz0^2\sqrt{x0^2+y0^2+z0^2}\right)\Big)\Big),$$

$$\frac{1}{4\left(x0^2+y0^2+z0^2\right)^4}\Big(4\,y0\left(x0^2+y0^2+z0^2\right)+24\,vy0\left(vx0\,x0+vy0\,y0+vz0\,z0\right)\left(x0^2+y0^2+z0^2\right)^{3/2}-3\,y0\Big(20\left(vx0\,x0+vy0\,y0+vz0\,z0\right)^2\sqrt{x0^2+y0^2+z0^2}-4\left(x0^2+y0^2+z0^2\right)\left(-1+vx0^2\sqrt{x0^2+y0^2+z0^2}+vy0^2\sqrt{x0^2+y0^2+z0^2}+vz0^2\sqrt{x0^2+y0^2+z0^2}\right)\Big)\Big),$$

$$\frac{1}{4\left(x0^2+y0^2+z0^2\right)^4}\Big(4\,z0\left(x0^2+y0^2+z0^2\right)+24\,vz0\left(vx0\,x0+vy0\,y0+vz0\,z0\right)\left(x0^2+y0^2+z0^2\right)^{3/2}-3\,z0\Big(20\left(vx0\,x0+vy0\,y0+vz0\,z0\right)^2\sqrt{x0^2+y0^2+z0^2}-4\left(x0^2+y0^2+z0^2\right)\left(-1+vx0^2\sqrt{x0^2+y0^2+z0^2}+vy0^2\sqrt{x0^2+y0^2+z0^2}+vz0^2\sqrt{x0^2+y0^2+z0^2}\right)\Big)\Big)\Big\}$$

Note: the complicated expression is copied and pasted from wA[5] into the function definition, NOT retyped.

In[170]:=
```
Clear[d4rvecBruteForce, d4rvec, d4rvecSLT, CompareCalculations];
d4rvecBruteForce[x0_, y0_, z0_, vx0_, vy0_, vz0_] :=
```

$$\Big\{\frac{1}{\left(x0^2+y0^2+z0^2\right)^{7/2}}\Big(3\left(-2\,vx0^2+vy0^2+vz0^2\right)x0^3-24\,vx0\,x0^2\left(vy0\,y0+vz0\,z0\right)+6\,vx0\left(vy0\,y0+vz0\,z0\right)\left(y0^2+z0^2\right)+x0\Big(3\left(3\,vx0^2-4\,vy0^2+vz0^2\right)y0^2-30\,vy0\,vz0\,y0\,z0+3\left(3\,vx0^2+vy0^2-4\,vz0^2\right)z0^2-2\sqrt{x0^2+y0^2+z0^2}\Big)\Big),$$

$$\frac{1}{\left(x0^2+y0^2+z0^2\right)^{7/2}}\Big(3\left(vx0^2-2\,vy0^2+vz0^2\right)y0^3-24\,vy0\,y0^2\left(vx0\,x0+vz0\,z0\right)+6\,vy0\left(vx0\,x0+vz0\,z0\right)\left(x0^2+z0^2\right)+y0\Big(3\left(-4\,vx0^2+3\,vy0^2+vz0^2\right)x0^2-30\,vx0\,vz0\,x0\,z0+3\left(vx0^2+3\,vy0^2-4\,vz0^2\right)z0^2-2\sqrt{x0^2+y0^2+z0^2}\Big)\Big),$$

$$\frac{1}{\left(x0^2+y0^2+z0^2\right)^{7/2}}\Big(6\,vz0\left(vx0\,x0+vy0\,y0\right)\left(x0^2+y0^2\right)-24\,vz0\left(vx0\,x0+vy0\,y0\right)z0^2+3\left(vx0^2+vy0^2-2\,vz0^2\right)z0^3+z0\Big(3\left(-4\,vx0^2+vy0^2+3\,vz0^2\right)x0^2-30\,vx0\,vy0\,x0\,y0+3\left(vx0^2-4\,vy0^2+3\,vz0^2\right)y0^2-2\sqrt{x0^2+y0^2+z0^2}\Big)\Big)\Big\};$$

```
d4rvec[x0_, y0_, z0_, vx0_, vy0_, vz0_] :=
  Module[{rvec0, vvec0, r0, v0, s0, sdot0, μ, σ, ϵ, f4, g4, d4r},
   rvec0 = {x0, y0, z0};
   vvec0 = {vx0, vy0, vz0};
   r0 = Dot[rvec0, rvec0];
   s0 = Dot[rvec0, vvec0];
   sdot0 = Dot[vvec0, vvec0];
   μ = 1 / r0^3;
   σ = s0 / r0^2;
   ϵ = sdot0 / r0^2;
   f4 = 3 ϵ μ - 2 μ^2 - 15 μ σ^2;
   g4 = 6 μ σ;
   d4r = f4 rvec0 + g4 vvec0];

d4rvecSLT[x0_, y0_, z0_, vx0_, vy0_, vz0_] :=
  Module[{rvec0, vvec0, r0, v0, s0, sdot0, μ, σ, ϵ, f4, g4, d4r},
   rvec0 = {x0, y0, z0};
   vvec0 = {vx0, vy0, vz0};
   r0 = Dot[rvec0, rvec0];
   s0 = Dot[rvec0, vvec0];
   sdot0 = Dot[vvec0, vvec0];
   μ = 1 / r0^3;
   σ = s0 / r0^2;
   ϵ = sdot0 / r0^2;
   f4 = 3 ϵ μ + μ^2 - 15 μ σ^2;
   g4 = 6 μ σ;
   d4r = f4 rvec0 + g4 vvec0];

CompareCalculations[x0_, y0_, z0_, vx0_, vy0_, vz0_] :=
  {d4rvec[x0, y0, z0, vx0, vy0, vz0],
   d4rvecSLT[x0, y0, z0, vx0, vy0, vz0], d4rvecBruteForce[x0, y0, z0, vx0, vy0, vz0],
   d4rvec[x0, y0, z0, vx0, vy0, vz0] - d4rvecBruteForce[x0, y0, z0, vx0, vy0, vz0]};
```